

Functional Pearl

A Greedy Algorithm for Dropping Digits

Richard Bird¹ Shin-Cheng Mu²

¹Department of Computer Science, University of Oxford

²Institute of Information Science, Academia Sinica

Abstract

Consider the puzzle: given a number, remove k digits such that the resulting number is as large as possible. Various techniques were employed to derive a linear-time solution to the puzzle: predicate logic was used to justify the structure of a greedy algorithm, a dependently-typed proof assistant was used to give a constructive proof of the greedy condition, and equational reasoning was used to calculate the greedy step as well as the final, linear-time optimisation.

1 INTRODUCTION

Greedy algorithms abound in computing. Well-known examples include Huffman coding, minimum cost spanning trees, and the coin-changing problem; but there are many others. This pearl adds yet another problem to this collection. However, as has been said before, greedy algorithms can be tricky things. The trickiness is not in the algorithm itself, which is usually quite short and easy to understand, but in the proof that it does produce a best possible result. The mathematical theory of *matroids*, see Lawler [1976], and its generalisation to *greedoids*, see Korte et al. [1991], have been developed to explain why and when many greedy algorithms work, although the theory does not cover all possible cases. In practice, greedy algorithms are usually verified directly rather than by extracting the underlying matroid or greedoid. Curtis [2003] discusses four basic ways in which a greedy algorithm can be proved to work, one of which will be followed with our problem.

The problem is to remove k digits from a number containing at least k digits, so that the result is as large as possible. For example, removing one digit from the number "6782334" gives "782334" as the largest possible result, while removing three digit yields "8334". Given that a number can be seen as a list of digits, the problem can be generalised to removing, from a list whose elements are drawn from a linearly ordered type, k elements so that the result is largest under lexicographic ordering. While the problem was invented out of curiosity rather than for a pressing application, it has apparently been used as an interview question for candidates seeking jobs in computing.¹ The hope is that we can discover an algorithm that takes linear time in the number of elements.

¹ The problem is listed on LeetCode as <https://leetcode.com/problems/remove-k-digits/>, where the objective is to find the smallest number rather than the largest, but the principles are the same.

The first task is to give a formal specification of the problem. Consider the function *drops* that removes a single element from a non-empty list in all possible ways: ²

$$\begin{aligned} \text{drops} &:: \text{List } a \rightarrow \text{List } (\text{List } a) \\ \text{drops } [x] &= [[]] \\ \text{drops } (x : xs) &= xs : \text{map } (x:) (\text{drops } xs) \end{aligned}$$

For example, *drops* "abcd" = ["bcd", "acd", "abd", "abc"]. The function *solve* for computing a solution can be defined by a simple exhaustive search:

$$\begin{aligned} \text{solve} &:: \text{Ord } a \Rightarrow \text{Nat} \rightarrow \text{List } a \rightarrow \text{List } a \\ \text{solve } k &= \text{maximum} \cdot \text{apply } k \text{ step} \cdot \text{wrap} \text{ ,} \\ \text{step} &:: \text{List } (\text{List } a) \rightarrow \text{List } (\text{List } a) \\ \text{step} &= \text{concat} \cdot \text{map } \text{drops} \text{ .} \end{aligned}$$

The function *solve* converts a given input into a singleton list, applies the function *step* exactly *k* times to produce all possible candidates, and computes the lexical maximum of the result. *Nat* is the type of natural numbers. The function *step* is *drops* lifted to a list of candidates. It computes, for each candidate, all the ways to drop 1 element. Functions *wrap* and *apply* are respectively defined by

$$\begin{aligned} \text{wrap} &:: a \rightarrow \text{List } a & \text{apply} &:: \text{Nat} \rightarrow (a \rightarrow a) \rightarrow a \rightarrow a \\ \text{wrap } x &= [x] \text{ ,} & \text{apply } 0 & \quad f = \text{id} \\ & & \text{apply } (1 + k) & f = \text{apply } k f \cdot f \text{ ,} \end{aligned}$$

For brevity, for the rest of the pearl we will write *apply k f* as f^k . Since a sequence of length *n* has *n* drops, and computing the larger of two lists of length *n - k* takes $O(n - k)$ steps, this method for computing the answer takes $O(n^k)$ steps. We aim to do better.

2 A GREEDY ALGORITHM

To obtain a greedy algorithm, one would wish that the best way to remove *k* digits can be computed by removing 1 digit *k* times, and each time we greedily remove the digit that makes the current result as large as possible. That is, letting

$$\begin{aligned} \text{gstep} &:: \text{Ord } a \Rightarrow \text{List } a \rightarrow \text{List } a \\ \text{gstep} &= \text{maximum} \cdot \text{drops} \text{ ,} \end{aligned}$$

we wish to have

$$\text{maximum} \cdot \text{step}^k \cdot \text{wrap} = \text{gstep}^k \text{ .} \quad (1)$$

One cannot just claim that this strategy works without proper reasoning, however. It can be shown that (1) is true if the following monotonicity condition holds (we denote lexicographic ordering on lists by \trianglelefteq), and ordering on individual elements by \leq):

$$xs \trianglelefteq ys \Rightarrow (\forall xs' \in \text{drops } xs : (\exists ys' \in \text{drops } ys : xs' \trianglelefteq ys')) \text{ ,} \quad (2)$$

² We use notations similar to Haskell, with slight variations. For example, the type for lists is denoted by *List*, and we allow $(1 + k)$ as a pattern in function definitions, to match our inductive proofs. Laziness is not needed.

where (\in) is overloaded to denote membership for lists. That is, if ys is no worse than xs , whatever candidate we can obtain from xs , we can compute a candidate from ys that is no worse either.

Unfortunately, (2) does not hold for our problem. Consider $xs = "1934" \triangleleft "4234" = ys$, "934" is a possible result of *drops* xs , but the best we can do by removing one digit from ys is "434". Note that (2) does not hold even if we restrict xs and ys to lists that can be obtained from the same source — certainly both "1934" and "4234" are both result of removing two digits from, say, "194234".

In the terminology of Curtis [2003], the *Better-Global* principle — which says that if one first step is no worse than another, then there is a global solution using the former that is no worse than one using the latter — does not hold for this problem. What does hold is a weaker property, the *Best-Global* principle: a globally optimal solution can be obtained by starting out with the *best* possible step. Formally, what we do have is that for all xs and k :

$$\begin{aligned} (\forall xs' \in \text{step}^{1+k} [xs]: \\ (\exists zs \in (\text{step}^k \cdot \text{wrap} \cdot \text{gstep}) xs : xs' \trianglelefteq zs)) . \end{aligned} \quad (3)$$

That is, letting xs' be an arbitrary result of dropping $k+1$ elements from xs , one can always obtain a result zs that is no worse than xs' by greedily dropping the best element (by *gstep*) and then dropping arbitrary k elements.

Property (3) will be proved in Section 4. For now, let us see how (3) helps to prove (1), that is, $\text{maximum} \cdot \text{step}^k \cdot \text{wrap} = \text{gstep}^k$. The proof proceeds by induction on k . For $k := 0$ both sides reduce to *id*. For the inductive case we need the universal property of *maximum*: for all $s :: a \rightarrow b$ and $p :: a \rightarrow \text{List } b$, and for total order (\trianglelefteq) on b :³

$$\begin{aligned} s = \text{maximum} \cdot p \equiv & (\forall x : s \ x \in p \ x) \wedge \\ & (\forall x, y : y \in p \ x \Rightarrow y \trianglelefteq s \ x) . \end{aligned}$$

To prove $\text{maximum} \cdot \text{step}^{1+k} \cdot \text{wrap} = \text{gstep}^{1+k}$ we need to show that 1. for all xs , $\text{gstep}^{1+k} xs$ is a member of $\text{step}^{1+k} [xs]$, which is a routine proof, and 2. for all xs and for all $ys \in \text{step}^{1+k} [xs]$, we have $ys \trianglelefteq \text{gstep}^{1+k} xs$. We reason:

$$\begin{aligned} & ys \trianglelefteq \text{gstep}^{1+k} xs \\ \equiv & ys \trianglelefteq \text{gstep}^k (\text{gstep } xs) \\ \equiv & \{ \text{induction hypothesis} \} \\ & ys \trianglelefteq \text{maximum} (\text{step}^k [\text{gstep } xs]) \\ \equiv & \{ \text{since } ys \trianglelefteq \text{maximum } xss \equiv (\exists zs \in xss : ys \trianglelefteq zs) \} \\ & (\exists xs \in \text{step}^k [\text{gstep } xs] : ys \trianglelefteq xs) \\ \Leftarrow & \{ \text{by (3)} \} \\ & ys \in \text{step}^{1+k} [xs] , \end{aligned}$$

which is our assumption. We have thus proved (1).

Remark: the proof above was carried out using predicate logic. There is a relational counterpart, in the style of Bird and de Moor [1997], that is slightly more concise and more general, which we unfortunately cannot present without adding a section introducing the notations and rules.

³ We restrict our discussion to total orders to ensure that *maximum* returns one unique result. More general scenarios are discussed in Bird and de Moor [1997].

3 REFINING THE GREEDY STEP

We will prove the greedy condition (3) in Section 4. It will turn out that the proof makes use of properties of *gstep* that will be evident from its inductive definition. Therefore we calculate an inductive definition of *gstep* in this section.

It is easy to derive $\text{gstep } [x] = []$. For the inductive case we reason

$$\begin{aligned}
& \text{gstep } (x : y : xs) \\
&= \{ \text{definition of gstep} \} \\
& \quad \text{maximum } (\text{drops } (x : y : xs)) \\
&= \{ \text{definition of drops} \} \\
& \quad \text{maximum } ((y : xs) : \text{map } (x:) (\text{drops } (y : xs))) \\
&= \{ \text{definition of maximum} \} \\
& \quad \text{max } (y : xs) (\text{maximum } (\text{map } (x:) (\text{drops } (y : xs)))) \\
&= \{ \text{since maximum } (\text{map } (x:) xss) = x : \text{maximum } xss, \text{ provided } xss \text{ is nonempty} \} \\
& \quad \text{max } (y : xs) (x : \text{maximum } (\text{drops } (y : xs))) \\
&= \{ \text{definition of gstep} \} \\
& \quad \text{max } (y : xs) (x : \text{gstep } (y : xs)) \\
&= \{ \text{definition of max and lexicographic ordering} \} \\
& \quad \text{if } x < y \text{ then } y : xs \\
& \quad \quad \text{else if } x = y \text{ then } x : \text{max } xs (\text{gstep } (y : xs)) \\
& \quad \quad \quad \text{else } x : \text{gstep } (y : xs) \\
&= \{ \text{since } xs \trianglelefteq \text{gstep } (y : xs) \} \\
& \quad \text{if } x < y \text{ then } y : xs \text{ else } x : \text{gstep } (y : xs) .
\end{aligned}$$

Hence we have

$$\begin{aligned}
\text{gstep } [x] &= [] \\
\text{gstep } (x : y : xs) &= \text{if } x < y \text{ then } y : xs \text{ else } x : \text{gstep } (y : xs) .
\end{aligned}$$

It turns out that *gstep xs* deletes *the last element of the longest descending prefix of xs*. For easy reference, we will refer to this element as the *hill foot* of the list. For example, $\text{gstep } "8766678" = "876678"$, where the hill foot, the element deleted, is the third 6.

4 PROVING THE GREEDY CONDITION

In this section we aim to prove (3), recited here:

$$(\forall xs' \in \text{step}^{1+k} [xs] : (\exists zs \in (\text{step}^k \cdot \text{wrap} \cdot \text{gstep}) xs : xs' \trianglelefteq zs)) .$$

Proving a proposition containing universal and existential quantification can be thought of as playing a game. The opponent challenges us by providing *xs* and a way of removing $1 + k$ elements to obtain *xs'*. We win by presenting a way of removing k elements from *gstep xs*, such that the result *zs* satisfies $xs' \trianglelefteq zs$. Equivalently, we present a way of removing $1 + k$ elements from *xs*, while making sure that the hill foot of *xs* is among the $1 + k$ elements removed. To prove (3) is to come up with a strategy to always win the game.

We could just invent the strategy and argue for its correctness. However, we experimented with another approach: could a proof assistant offer some help? Can we conjecture the existence of a function that, given the opponent's input, computes *zs*, and try to develop the function and the proof that $xs' \trianglelefteq zs$ at the same time, letting their developments mutually guide each other? It would be a modern realisation of Dijkstra's belief that a program and its proof should be developed hand-in-hand [Dijkstra, 1974].

THE DATATYPES. We modeled the problem in the dependently typed language/proof assistant Agda. For the setting-up, we need to define a number of datatypes. Firstly, given a type a with a total ordering (\leq) (which derives a strict ordering ($<$)), lexicographic ordering for $List\ a$ is defined by:⁴

```

data  $\leq$  ::  $List\ a \rightarrow List\ a \rightarrow Set$  where
  []  $\leq$  []
   $x < y \rightarrow (x : xs) \leq (y : ys)$ 
   $xs \leq ys \rightarrow (x : xs) \leq (x : ys)$  ,

```

that is, $[]$ is no larger than any lists, $(x : xs) \leq (y : ys)$ if $x < y$, and two lists having the same head is compared by their tails.

Secondly, rather than actually deleting elements of a list, in proofs it helps to remember which elements are deleted. The following datatype $Dels\ k\ xs$ can be seen as instructions on how k elements are deleted from xs :

```

data  $Dels$  ::  $Nat \rightarrow List\ a \rightarrow Set$  where
  end ::  $Dels\ 0\ []$ 
  keep ::  $Dels\ k\ xs \rightarrow Dels\ k\ (x : xs)$ 
  del ::  $Dels\ k\ xs \rightarrow Dels\ (1 + k)\ (x : xs)$  .

```

For example, letting $xs = "abcde"$, $ds = keep\ (del\ (keep\ (del\ (keep\ end))))$:: $Dels\ 2\ xs$ says that the 1st and 3rd elements of xs (counting from 0) are to be deleted. The function $dels$ actually carries out the instruction:

```

dels :: ( $xs :: List\ a$ )  $\rightarrow Dels\ k\ xs \rightarrow List\ a$ 
dels [] end = []
dels (x : xs) (del ds) = dels xs ds
dels (x : xs) (keep ds) = x : dels xs ds .

```

For example, $dels\ xs\ ds = "ace"$.

Thirdly, the predicate $HFoot\ i\ xs$ holds if the i -th element in xs is the hill foot, that is, the element that would be removed by $gstep\ xs$:

```

data  $HFoot$  ::  $Nat \rightarrow List\ a \rightarrow Set$  where
  last ::  $HFoot\ 0\ (x : [])$ 
  this ::  $x < y \rightarrow HFoot\ 0\ (x : y : xs)$ 
  next ::  $x \geq y \rightarrow HFoot\ i\ (y : xs) \rightarrow HFoot\ (1 + i)\ (x : y : xs)$  .

```

For example, $next\ (next\ (next\ (next\ this)))$ may have type $HFoot\ 4\ "8766678"$, since the 4th element is the last in the descending prefix "87666".

Finally, we define a datatype $IsDel :: Nat \rightarrow Dels\ k\ xs \rightarrow Set$ such that, for all $ds : Dels\ k\ xs$, the relation $IsDel\ i\ ds$ holds if ds instructs that the i -th element of xs is to be deleted. Its definition is repetitive and thus omitted.

THE FUNCTION AND THE PROOFS. The aim is to construct the following function $alter$:

```

alter ::  $Dels\ (1 + k)\ xs \rightarrow HFoot\ i\ xs \rightarrow Dels\ (1 + k)\ xs$  .

```

It takes an instruction, given by the opponent, that deletes $1 + k$ elements from xs , and an evidence that the i -th element of xs is its hill foot, and produces a possibly altered instruction that also deletes $1 + k$ elements.

⁴ To be consistent with earlier parts of this pearl, we use Haskell-like notations for the Agda code. Typing relation is denote by $(::)$ and list cons by $(:)$. The two constructors of Nat are 0 and $(1+)$. Universally quantified implicit arguments in constructor and function declarations are omitted.

Recalling the discussion in the beginning of this section, *alter* should satisfy two properties:

$$\begin{aligned} \text{mono} &:: (ds :: \text{Dels } (1 + k) \text{ } xs) \rightarrow (ft :: \text{HFoot } i \text{ } xs) \rightarrow \text{dels } xs \text{ } ds \leq \text{dels } xs \text{ } (\text{alter } ds \text{ } ft) \text{ ,} \\ \text{unfoot} &:: (ds :: \text{Dels } (1 + k) \text{ } xs) \rightarrow (ft :: \text{HFoot } i \text{ } xs) \rightarrow \text{IsDel } i \text{ } (\text{alter } ds \text{ } ft) \text{ .} \end{aligned}$$

Given *ds* and *ft*, the property *mono* says that *alter ds ft* always produces a list that is not worse than that produced by *ds*, while *unfoot* says that *alter ds ft* does delete the hill foot.

The goal now is to develop *alter*, *mono*, and *unfoot* together. The reader is invited to give it a try — it is more fun trying it yourself! ⁵ In most of the steps, the type and proof constraints leave us with only one reasonable choice, while in one case we are led to discover a lemma. The cases to consider are:⁶

1. *alter* {*xs* = *x* : *y* : *ys*} (*keep ds*) (*this x < y*) — the opponent keeps *x*, which is the hill foot because $x < y$. Due to *unfoot*, we have to delete *x*; a simple way to satisfy *mono* is to keep *y*. Thus we return *del (keep ds')*, where *ds'* can be any instruction that deletes *k* elements in *xs* — it doesn't matter how *ds'* does it!
2. *alter* {*xs* = *x* : *y* : *ys*} (*keep ds*) (*next x ≥ y ft*) — the opponent keeps *x*, and we have not reached the hill foot yet. In this case it is safe to imitate the opponent and keep *x* too, before recursively calling *alter* to generate the rest of the instruction.
3. *alter* {*xs* = [*x*]} (*del end*) *last* — the opponent deletes the sole element *x*. In this case we delete *x* too, returning *del end*.
4. *alter* {*xs* = *x* : *y* : *ys*} (*del ds*) (*this x < y*) — the element *x* is the hill foot, and is deleted by the opponent. In this case, since *unfoot* is satisfied, we can do exactly the same. We end up returning the same instruction as the opponent's but it is fine, since both *mono* and *unfoot* are satisfied.
5. *alter* {*xs* = *x* : *y* : *ys*} (*del ds*) (*next x ≥ y ft*) — the opponent deletes *x*, which is in the descending prefix but is not the hill foot. This turns out to be the most complex case. One may try to imitate and delete *x* as well, returning *del ds'* for some *ds'*. However, *ds'*, having type *Dels k (y : ys)*, cannot be produced by a recursive call to *alter*, whose return type is *Dels (1 + k) _*. It could be the case that *k* is 0 and, since we have not deleted the hill foot yet, returning *Dels 0 (y : ys)* would violate *unfoot*. The lesson learnt from the type is that we can only delete $1 + k$ elements, and we have to save at least one *del* for the hill foot, which is yet to come. We thus have to further distinguish between two cases:
 - (a) $k = 1 + k'$ for some k' . In this case we still have room to delete more elements, thus we can safely imitate the opponent, delete *x*, and recursively call *alter*.
 - (b) $k = 0$. In this case we keep *x*, returning *keep (delfoot ft) :: Dels 1 (y : ys)*, where *delfoot :: HFoot i zs → Dels 1 zs* computes an instruction

⁵ The Agda code can be downloaded from <https://scm.iis.sinica.edu.tw/home/2020/dropping-digits/>.

⁶ Curly brackets are used in Agda to mention implicit arguments. In each case here we pattern match {*xs* = ...} such that the readers know what input list we are dealing with.

```

alter :: Dels (1 + k) xs → HFoot i xs → Dels (1 + k) xs
alter (keep ds) (this x < y) = del (keep (deleteAny ds))
alter (keep ds) (next x ≥ y ft) = keep (alter ds ft)
alter (del end) last = del end
alter (del ds) (this x < y) = del ds
alter {k = 0} (del ds) (next x ≥ y ft) = keep (delfoot ft)
alter {k = 1 + k} (del ds) (next x ≥ y ft) = del (alter ds ft) ,

```




Figure 1: The function `alter`, where `deleteAny ds` generates a `Dels k xs`, and a graphical summary. Element with dotted outline are those that are considered already; the one with thick outline is the current element. It is the hill foot if it is smaller than the element to the right, or it is the last. Deleted elements are marked with a cross.

that deletes exactly one element, the hill foot. What is left to prove to establish `mono` for this case can be extracted to be a lemma:

$$\begin{aligned}
\text{monoAux} :: x \geq y \rightarrow (ft :: \text{HFoot } i (y : ys)) \rightarrow \\
(y : ys) \leq (x : \text{dels } (y : ys) (\text{delfoot } ft)) ,
\end{aligned}$$

whose proof is an induction on `ys`, keeping `x ≥ y` as an invariant. If `x < y` or `y` is the hill foot, we are done. Otherwise `x = y` and we inductively inspect the tail `ys`. Without Agda, it would not be easy to discover this lemma.

In summary, the function `alter` we have constructed, and a graphical summary, are shown in Figure 1.

Remark: We may also tuple `alter` and the properties together, and try to construct:

$$\begin{aligned}
\text{alter}' :: (ds :: \text{Dels } (1 + k) xs) \rightarrow (ft :: \text{HFoot } i xs) \rightarrow \\
\exists (\lambda(ds' :: \text{Dels } (1 + k) xs) \rightarrow (\text{dels } xs ds \leq \text{dels } xs ds') \times \text{IsDel } i ds') .
\end{aligned}$$

An advantage is that the code of each case of `alter` are next to its proof. A disadvantage is that having to pattern-match the result of `alter'` psychologically discourages one from making a recursive call when needing a `Dels (1 + k)` ... It is up to personal preference which style one prefers.

5 IMPROVING EFFICIENCY

Back to our code. We have proved that `solve k = gstepk`, with `gstep` given by:

$$\begin{aligned} gstep [x] &= [] \\ gstep (x : y : xs) &= \text{if } x < y \text{ then } y : xs \text{ else } x : gstep (y : xs) . \end{aligned}$$

Each time *gstep* is called, it takes $O(n)$ steps to go through the descending prefix and find the hill foot, before the next invocation of *gstep* starts from the beginning of the list again. Therefore, *solve k* takes $O(kn)$ steps over all. This is certainly not necessary — to find the next hill foot, the next *gstep* could start from where the previous one left off.

The way to implement this idea is to bring in an accumulating parameter. Suppose we generalise *solve* to a function *gsolve*, defined by

$$gsolve k xs ys = solve k (xs ++ ys) ,$$

with the proviso that the argument *xs* is constrained to be a descending sequence. In particular, $solve k xs = gsolve k [] xs$. We aim to develop a recursive definition of *gsolve*. Clearly,

$$gsolve 0 xs ys = xs ++ ys .$$

Recalling that *gstep* drops the last element of a descending list, we know that *k* repetitions of *gstep* on a decreasing list will drop the last *k* elements. Hence

$$gsolve k xs [] = dropLast k xs ,$$

where *dropLast k* drops the last *k* elements of a list. We will not give a formal definition of *dropLast* as it will be replaced by another function in a moment. That deals with the two base cases. For the recursive case, it is easy to prove the following property of *gstep*:

$$\begin{aligned} gstep (xs ++ y : ys) \mid \text{null } xs \vee \text{last } xs \geq y &= gstep ((xs ++ [y]) ++ ys) \\ \mid \text{otherwise} &= \text{init } xs ++ y : ys , \end{aligned}$$

which can be used to construct the following case of *gsolve*:

$$\begin{aligned} gsolve (1 + k) xs (y : ys) \mid \text{null } xs \vee \text{last } xs \geq y &= gsolve (1 + k) (xs ++ [y]) ys \\ \mid \text{otherwise} &= gsolve k (\text{init } xs) (y : ys) . \end{aligned}$$

The second optimisation is simply to replace the list *xs* in the definition of *gsolve* by *reverse xs* to avoid adding elements at the end of a list. That leads to our final algorithm:

$$\begin{aligned} solve k xs &= gsolve k [] xs , \\ gsolve 0 xs ys &= reverse xs ++ ys \\ gsolve k xs [] &= reverse (drop k xs) \\ gsolve k xs (y : ys) \mid \text{null } xs \vee \text{head } xs \geq y &= gsolve k (y : xs) ys \\ \mid \text{otherwise} &= gsolve (k - 1) (\text{tail } xs) (y : ys) , \end{aligned}$$

where *drop k* is a standard Haskell function that drops the first *k* elements from a list. For an operational explanation, *gsolve* traverses through the list, keeping looking for the next hill foot to delete — the *otherwise* case is when a hill foot is found. The list *xs* is the traversed part — (xs, ys) forms a zipper. The head of *xs* is a possible candidate of the hill foot. While the algorithm looks simple once understood, without calculation it is not easy to get the details right. The authors have come up with several versions that are wrong, before sitting down to calculate it!

To time the program, note that at each step either *k* is reduced to *k - 1* or *y : ys* is reduced to *ys*. Hence *solve k xs* takes $O(k + n) = O(n)$ steps, where $n = \text{length } xs$.

6 CONCLUSION

To construct a linear-time algorithm for solving the puzzle, various techniques were employed. The structure of the greedy algorithm was proved using predicate logic, and the proof was simplified from relational program calculus. Agda was used to give a constructive proof of the greedy condition, and equational reasoning was used to derive the greedy step as well as the final, linear-time optimisation.

REFERENCES

- R. S. Bird and O. de Moor. *Algebra of Programming*. International Series in Computer Science. Prentice Hall, 1997. ISBN 0-13-507245-X.
- S. Curtis. The classification of greedy algorithms. *Science of Computer Programming*, 49(1-3):125–157, 2003.
- E. W. Dijkstra. Programming as a discipline of mathematical nature. *American Mathematical Monthly*, 81(6):608–612, May 1974. EWD 361.
- B. Korte, L. Lovász, and R. Schrader. *Greedoids*. Springer-Verlag, 1991.
- E. L. Lawler. *Combinatorial Optimization: Networks and Matroids*. Holt, Rinehart, and Winston, 1976.